Programación Funcional Avanzada

Programación Dinámica Funcional

Ernesto Hernández-Novich
<emhn@usb.ve>

Universidad "Simón Bolívar"

UNIVERSIDAD SIMÓN BOLÍVAR

Programación Dinámica

Calcular una vez, compartir varias

- Convertir un problema de programación en una recurrencia – calcularla almacenando valores intermedios en una colección.
- En lenguajes imperativos la colección es un arreglo – intercambiar espacio por tiempo eliminando redundancia.
- Calcular una vez y compartirlo varias veces – *exactamente* lo que hace la evaluación perezosa.

UNIVERSIDAD SIMÓN BOLÍVAR

Se venden en combos...

- Se pueden comprar tequeños en combos de 6, 9 y 20.
- Se pueden comprar cero o más combos.

¿Se pueden comprar *exactamente* `N` tequeños?

¿Puedo comprar *exactamente* N?

Como un problema de decisión

```
comprar0 n = r!n
    where r   = listArray (0,n)
                          (True : map f [1..n])
          f i = i >=  6 && r!(i-6)  ||
                i >=  9 && r!(i-9)  ||
                i >= 20 && r!(i-20)
```

- n–ésima posición del arreglo `r` indica si se puede o no.
- Si puedo comprar $i-6$ o $i-9$ o $i-20$ tequeños, entonces puedo comprar i.
- La posición inicial es `True`.
- Arreglo *inmutable* se genera con una iteración cerrada.

UNIVERSIDAD SIMÓN BOLÍVAR

¿Puedo comprar *exactamente* N?

Como un problema de decisión

```
comprar0 n = r!n
    where r   = listArray (0,n)
                          (True : map f [1..n])
          f i = i >=  6 && r!(i-6)  ||
                i >=  9 && r!(i-9)  ||
                i >= 20 && r!(i-20)
```

- n—ésima posición del arreglo `r` indica si se puede o no.
- Si puedo comprar $i-6$ o $i-9$ o $i-20$ tequeños, entonces puedo comprar i.
- La posición inicial es `True`.
- Arreglo *inmutable* se genera con una iteración cerrada.

Si pero, ¿*cuántos* combos necesito?

UNIVERSIDAD SIMÓN BOLÍVAR

¿Puedo comprar *exactamente* N?

Como un problema de combinación

```
comprar1 n = r!n
  where r = listArray (0,n)
                      (Just (0,0,0) : map f [1..n])
        f i = case attempt (i-6) of
          Just (x,y,z) -> Just (x+1,y,z)
          _            -> case attempt (i-9) of
             Just (x,y,z) -> Just (x,y+1,z)
             _            -> case attempt (i-20) of
                Just (x,y,z) -> Just (x,y,z+1)
                _            -> Nothing
        attempt x = if x>=0 then r!x
                            else Nothing
```

- Arreglo indica `Maybe (Int,Int,Int)` – cuántos combos de 6, 9 y 20.

UNIVERSIDAD SIMÓN BOLÍVAR

¿Puedo comprar *exactamente* N?

Como un problema de combinación

```
comprar1 n = r!n
  where r = listArray (0,n)
                      (Just (0,0,0) : map f [1..n])
        f i = case attempt (i-6) of
          Just (x,y,z) -> Just (x+1,y,z)
          _            -> case attempt (i-9) of
             Just (x,y,z) -> Just (x,y+1,z)
             _            -> case attempt (i-20) of
                Just (x,y,z) -> Just (x,y,z+1)
                _            -> Nothing
        attempt x = if x>=0 then r!x
                            else Nothing
```

- Arreglo indica `Maybe (Int,Int,Int)` – cuántos combos de 6, 9 y 20.
- Esa cascada de case apesta a Monad `Maybe`.

UNIVERSIDAD SIMÓN BOLÍVAR

Una solución más regular e idiomática

```
comprarM n = r!n
  where r = listArray (0,n)
                      (Just (0,0,0) : map f [1..n])
        f i = do (x,y,z) <- attempt (i-6)
                 return (x+1,y,z)
               `mplus`
               do (x,y,z) <- attempt (i-9)
                  return (x,y+1,z)
               `mplus`
               do (x,y,z) <- attempt (i-20)
                  return (x,y,z+1)
        attempt x = guard (x>=0) >> r!x
```

- Maybe es Monad – attempt usa guard para "corto-circuito".
- Maybe es MonadPlus – mplus produce el primer resultado o Nothing

No estamos siendo muy funcionales...

- Estamos manteniendo *todo* el arreglo en memoria – ¿qué va a pasar cuando `n` sea grande?
- Sería más eficiente si pudiéramos *reciclar* las posiciones bajas del arreglo una vez que no hacen falta.
- Eso lo hace el recolector de basura si realizamos los cómputos de manera que los valores previos no hagan falta.

UNIVERSIDAD SIMÓN BOLÍVAR

Un viejo truco que siempre está vigente

Una lista que se consume a si misma

```
comprar2 n = go n (True : replicate 19 False)
  where go 0 cs = cs !! 0
        go n cs = go (n-1)
                     ((cs !! 5 ||
                       cs !! 8 ||
                       cs !! 19) : take 19 cs)
```

- Nunca necesitamos más de 20 elementos.
- El primero siempre es `True` y "sembramos" el resto con `False`.

UNIVERSIDAD SIMÓN BOLÍVAR

Y ahora bajamos de nivel...

```
comprar3 n = go n 1
  where
    go :: Int -> Int -> Bool
    go 0 cs = odd cs
    go n cs = go (n-1) ((cs `shiftL` 1) .|.
                        if cs .&. 0x80120 /= 0 then 1
                                               else 0)
```

- $80120_{16} = 10000000000100100000_2$ – bits 20, 9 y 6.
- El arreglo está empaquetado en un mapa de bits – `Data.Bits`

UNIVERSIDAD SIMÓN BOLÍVAR

¿Cuál es más rápido?

Usemos `Criterion`

Función	Promedio
comprar0	167.7562 μs
comprar1	192.1358 μs
comprarM	190.4110 μs
comprar2	663.5192 μs
comprar3	008.2330 μs

¿Cuál es más rápido?

Usemos `Criterion`

Función	Promedio
comprar0	167.7562 μs
comprar1	192.1358 μs
comprarM	190.4110 μs
comprar2	663.5192 μs
comprar3	008.2330 μs

¿Cuánto cuesta hacerlo así de rápido en C?

UNIVERSIDAD SIMÓN BOLÍVAR

Distancia de Levenshtein

Usado en procesamiento de lenguaje natural

- Medida de la diferencia entre dos secuencias – usualmente cadenas.
- Número mínimo de ediciones (agregar, eliminar o sustituir) para convertir una cadena en la otra.
- La distancia entre `maduro` y `balurdo` es tres
  - `Maduro` pasa a `Baduro` – sustitución.
  - `baDuro` pasa a `baLuro` – sustitucion.
  - `baluro` pasa a `balurDo` – inserción.
  - No hay manera de hacerlo en menos de tres ediciones.

¿Cómo se calcula?

UNIVERSIDAD SIMÓN BOLÍVAR

Distancia de Levenshtein

Una recurrencia

$$lev_{a,b}(i,j) = \begin{cases} max(i,j) & \text{cuando } min(i,j) = 0 \\ min & \begin{cases} lev_{a,b}(i-1,j)+1 \\ lev_{a,b}(i,j-1)+1 \\ lev_{a,b}(i-1,j-1)+[a_i \neq b_j] \end{cases} \end{cases}$$

- a y b son las cadenas a comparar – distancia es $lev_{a,b}(|a|,|b|)$
- La implantación recursiva es obvia e ineficiente.
- La implantación dinámica típica usa una matriz para conservar las distancias entre *todos* los prefijos.

UNIVERSIDAD SIMÓN BOLÍVAR

La matriz de “maduro” a “balurdo”

Guardando las distancias...

$$lev_{a,b}(i,j) = \begin{cases} max(i,j) & \text{cuando } min(i,j) = 0 \\ min & \begin{cases} lev_{a,b}(i-1,j)+1 \\ lev_{a,b}(i,j-1)+1 \\ lev_{a,b}(i-1,j-1)+[a_i \neq b_j] \end{cases} \end{cases}$$

		m	a	d	u	r	o
	0	1	2	3	4	5	6
b	1	1	2	3	4	5	6
a	2	2	1	2	3	4	5
l	3	3	2	2	3	4	5
u	4	4	3	3	2	3	4
r	5	5	4	4	3	2	3
d	6	6	5	4	4	3	3
o	7	7	6	5	5	4	3

UNIVERSIDAD SIMÓN BOLÍVAR

Expresado recursivamente

Se presta para listas por comprensión

```
lev0 s t = d !! (length s) !! (length t)
  where
    d = [
          [ delta m n | n <- [0 .. length t] ]
          | m <- [0 .. length s]
        ]
    delta i 0 = i
    delta 0 j = j
    delta i j = minimum [
        d !! (i-1) !! j + 1,
        d !! i     !! (j-1) + 1,
        d !! (i-1) !! (j-1) +
                      (if s!!(i-1)==t!!(j-1) then 0
                                             else 1) ]
```

Con arreglos ha de ser más rápido...

UNIVERSIDAD SIMÓN BOLÍVAR

Si tan sólo tuviera arreglos

- Función de índices a valores – nadie quiere escribir todos los casos...
- `Data.Ix` – noción generalizada de "tipo índice"

```
class Ord a => Ix a where
  range     :: (a, a) -> [a]
  index     :: (a, a) -> a -> Int
  inRange   :: (a, a) -> a -> Bool
  rangeSize :: (a, a) -> Int
```

  - Biyección entre tipo ordenado y los enteros.
  - Tamaño y verificación del rango explícitas.

Si algo es ordenable, puede servir de índice...

UNIVERSIDAD SIMÓN BOLÍVAR

Indices. . . te los tengo generalizados!

```
data MetaVar = Foo | Bar | Baz | Qux | Grok
             deriving (Eq,Ord,Ix,Show)

> range (Foo,Baz)
[Foo,Bar,Baz]
> index (Bar,Qux) Baz
1
> inRange (Foo,Qux) Grok
False
```

- Sirve cualquier sub-rango de un rango de valores ordenados.
- Si puedes ordenar cosas, puedes usarlas como índices.
- Números, enumeraciones, tuplas . . .

UNIVERSIDAD SIMÓN BOLÍVAR

Arreglos. . . te los tengo!

Representación clásica

```
data Ix i => Array i e

array :: Ix i => (i, i) -> [(i, e)] -> Array i e
(!)   :: Ix i => Array i e -> i -> e
```

- `array` – construye a partir del rango y lista de elementos correspondientes.
- `(!)` – es el operador de subindizado.
- Estricto en los límites del rango y en los índices – perezoso en los valores almacenados en el arreglo.

UNIVERSIDAD SIMÓN BOLÍVAR

Arreglos. . . te los tengo!

Representación clásica

```
sqs = array (1,100) [(i, i*i) | i <- [1..100]]
nms = array (Foo,Baz) [(m,v) |
            m <- range (Foo,Baz),
            v <- [0..rangeSize (Foo,Baz) - 1]]

> sqs ! 7
49
> nms
array (Foo,Baz) [(Foo,2),(Bar,2),(Baz,2)]
> nms ! Bar
2
> sqs ! (nms ! Bar)
4
```

UNIVERSIDAD SIMÓN BOLÍVAR

Arreglos. . . te los tengo!

Definición recursiva y perezosa

```
fibs     :: Int -> Array Int Int
fibs n = a
  where a = array
              (0,n) $
              [(0, 1), (1, 1)] ++
              [(i, a!(i-2) + a!(i-1)) | i <- [2..n]]

> fibs 5
array (0,6) [(0,1),(1,1),(2,2),(3,3),(4,5),(5,8)]
> fibs ! 4
5
```

- El arreglo se define en función de si mismo.
- Muy útil en algunos problemas de programación dinámica.

UNIVERSIDAD SIMÓN BOLÍVAR

Arreglos. . . te los tengo!

¡En forma de matriz recursiva!

```
wavefront    :: Int -> Array (Int,Int) Int
wavefront n = a
 where a =
  array ((1,1),(n,n)) $
        [((1,j), 1) | j <- [1..n]] ++
        [((i,1), 1) | i <- [2..n]] ++
        [((i,j), a!(i,j-1) + a!(i-1,j-1) + a!(i-1,j))
                    | i <- [2..n], j <- [2..n]]
```

- Uso tuplas como índices – ¡en forma de matriz!
- Primera fila y primera columna llenas de unos.
- Cada celda es la suma de sus vecinos norte, oeste y noroeste.

UNIVERSIDAD SIMÓN BOLÍVAR

Por favor dime que son mutables

```
(//) :: (Ix a) => Array a b -> [(a,b)] -> Array a b
```

- Dado un arreglo *reemplazar* ciertas posiciones.
- Arreglos clásicos son monolíticos y puros.
  - Elementos individuales son inmutables.
  - `(//)` **copia** el arreglo incorporando los cambios.
- Suficientes si uno necesita una tabla multidimensional estática – insuficientes si uno necesita transformar la tabla progresivamente.

Por eso es que hay librerías alternativas...

UNIVERSIDAD SIMÓN BOLÍVAR

Una taxonomía de arreglos

. . . y sus repercusiones

- `Array` – puros e inmutables
- `STArray` – mónadicos mutables en el `Monad ST`
- `IOArray` – mónadicos mutables en el `Monad IO`

UNIVERSIDAD SIMÓN BOLÍVAR

Array – Arreglos Inmutables

Absolutamente puros

```
import Data.Array

change :: (Array Int Int, Array Int Int)
change = (original, cambiado)
   where original = listArray (1,4) (repeat 42)
         cambiado = original // [(2, 69)]

main = print change
```

- Excelentes para código puro con búsquedas al azar.
- Costosos para modificaciones – siempre se copian.
- `accumArray` – quizás no necesitabas mutarlo.
- `UArray` de `Data.Array.Unboxed` almacena valores *unboxed*.

UNIVERSIDAD SIMÓN BOLÍVAR

STArray – Arreglos Mutables en el Monad ST

Al borde de la impureza, pero libres de I/O

```
import Control.Monad.ST
import Data.Array.ST

change :: ST s (Int,Int)
change = do
  m <- newArray (1,4) 42 :: ST s (STArray s Int Int)
  a <- readArray m 2
  writeArray m 2 69
  b <- readArray m 2
  return (a,b)

main = print $ runST change
```

- Excelentes para código monádico con búsquedas al azar.
- Modificación económica – `Monad ST` mantiene el cambio.
- `STUArray` de `Data.Array.ST` almacena valores *unboxed*.

UNIVERSIDAD SIMÓN BOLÍVAR

IOArray – Arreglos Mutables en el Monad IO

Arreglos sin escapatoria de I/O

```
import Data.Array.IO

main = do
  m <- newArray (1,4) 42 :: IO (IOArray Int Int)
  a <- readArray m 2
  writeArray m 2 69
  b <- readArray m 2
  return (a,b)
```

- Cuando los contenidos van y vienen vía IO – aprovechar `hGetArray` y `hPutArray` para el intercambio.
- `IOUArray` de `Data.Array.IO` almacena valores *unboxed*.

UNIVERSIDAD SIMÓN BOLÍVAR

Ida y vuelta

De mutable a inmutable, y de regreso

```
freeze :: (Ix i, MArray a e m, IArray b e)
       => a i e -> m (b i e)
thaw   :: (Ix i, IArray a e, MArray b e m)
       => a i e -> m (b i e)
```

- `IArray` es la clase de arreglos inmutables – `Array`
- `MArray` es la clase de arreglos mutables – `IOArray` y `STArray`
- Ambas operaciones requieren sacar una copia completa del arrelgo.

Algunas condiciones aplican...

- Versiones *unboxed* sólo pueden almacenar valores con tamaño fijo.
  - Tipos primitivos van bien – `Bool`, `Int`, `Double`, `Char`, `Word`...
  - No puedes usar `Integer`, `String`, tuplas, ...
  - Como son *unboxed* deben evaluarse **todos** al intentar usar el arreglo – pierdes los beneficios de la evaluación perezosa.
  - A efectos prácticos son como los arreglos de C.
- Muchos arreglos mutables fastidian al recolector de basura – evítelos o use `+RTS -A8m -RTS` para aliviar el problema.

UNIVERSIDAD SIMÓN BOLÍVAR

Con arreglos ha de ser más rápido

- Imitar la solución imperativa requiere arreglos mutables.
  - Todos los cómputos son puros – operar en el Monad `ST`.
  - Datos *unboxed* para máximo desempeño.

Parece un trabajo para `STUArray`

UNIVERSIDAD SIMÓN BOLÍVAR

Arreglos mutables en Monad ST

La inicialización...

```
lev1 s t = d ! (ls , lt)
  where s' = array (0,ls)
                   [ (i,x) | (i,x) <- zip [0..] s ]
                   :: UArray Int Char
        t' = array (0,lt)
                   [ (i,x) | (i,x) <- zip [0..] t ]
                   :: UArray Int Char
        ls = length s
        lt = length t
        (l,h) = ((0,0),(length s,length t))
```

- `s'` y `t'` – acceso rápido al i–ésimo caracter de cada palabra.
- `l` y `h` – coordenadas mínima y máxima del arreglo mutable.

UNIVERSIDAD SIMÓN BOLÍVAR

Arreglos mutables en Monad ST

La transformación...

```
d = runSTUArray $ do
    m <- newArray (l,h) 0
          :: ST s (STUArray s (Int,Int) Int)
    forM_ [0..ls] $ \i -> writeArray m (i,0) i
    forM_ [0..lt] $ \j -> writeArray m (0,j) j
    forM_ [1..lt] $ \j -> do
      forM_ [1..ls] $ \i -> do
        let c = if s'!(i-1) == t'!(j-1) then 0 else 1
        x <- readArray m (i-1,j)
        y <- readArray m (i,j-1)
        z <- readArray m (i-1,j-1)
        writeArray m (i,j) $ minimum [x+1, y+1, z+c]
    return m
```

- Arreglo *unboxed* de dos dimensiones – no nos importa el estado.
- Se inicializan los "márgenes".
- Iteración monádica anidada – traducción del método iterativo.

UNIVERSIDAD SIMÓN BOLÍVAR

¿Y es más rápido?

Otra vez Criterion

```
benchmarking lev0
[...]
mean: 6.826855 us,
      lb 6.803381 us, ub 6.857277 us, ci 0.950

benchmarking lev1
[...]
mean: 2.909211 us,
      lb 2.901831 us, ub 2.917778 us, ci 0.950
```

Poco más del doble de rápido

Construyendo recursivamente

Reconsideremos el método...

- El arreglo se construye en **una** pasada.
- Cada celda se construye en base a vecinas previas –
  su valor no cambiará más adelante.
- En realidad, el arreglo es *inmutable* una vez construido.
- Sirve un vulgar (y puro) `Data.Array` –
  siempre y cuando se construya recursivamente.
- Cada celda es una *función* de sus vecinas –
  y la vecindad es una función de la coordenada actual.

```
mkArray f bs = array bs
                     [ (i, f i) | i <- range bs ]
```

UNIVERSIDAD SIMÓN BOLÍVAR

Construyendo recursivamente

Lazy functional epicness

```
lev2 sa sb = table ! (length sa, length sb)
  where
    arrA = listArray (0, length sa - 1) sa
    arrB = listArray (0, length sb - 1) sb
    table = mkArray f ((0,0), (length sa, length sb))
    f (ia, 0) = ia
    f (0 ,ib) = ib
    f (ia,ib)
      | a == b     = table ! (ia-1,ib-1)
      | otherwise = 1 + minimum [ table ! x |
                                  x <- [ (ia-1,ib-1),
                                         (ia-1,ib),
                                         (ia,ib-1)] ]
      where
        a = arrA ! (ia - 1)
        b = arrB ! (ib - 1)
```

¿Y es más rápido?

Otra vez Criterion

```
benchmarking lev0
[...]
mean: 6.826855 us,
      lb 6.803381 us, ub 6.857277 us, ci 0.950

benchmarking lev1
[...]
mean: 2.909211 us,
      lb 2.901831 us, ub 2.917778 us, ci 0.950

benchmarking lev2
[...]
mean: 2.222751 us,
      lb 2.216678 us, ub 2.230146 us, ci 0.950
```

Aún más rápido.

Quiero saber más...

- Documentación de `Data.Array`
- Documentación sobre Data.MemoTrie – Lazy Memoization using Tries
- Página sobre la Distancia de Levenshtein en WikiPedia

UNIVERSIDAD SIMÓN BOLÍVAR